Estimation of Direct, Period, and Carryover Effects in Crossover Studies

Mitchell L. Kotler, Colgate-Palmolive Company, Piscataway, NJ Larry L. Laster, University of Pennsylvania, Philadelphia, PA

Abstract

While the crossover design has been criticized in recent years, it remains a common method of testing differences among treatments with greater power than the simple parallel group design. We consider studies in which carryover (previous treatment) and period (time slot in which treatment was given) are effects on the response. The purpose of this paper is to describe procedures for analyzing data from such a crossover study (no missing data) using SAS®. Specifically, PROC GLM in SAS/STAT® is used, with ESTIMATE and LSMEANS statements to generate appropriate estimates. The assumptions underlying the analysis and the meaning of the estimates will be discussed. We also discuss alternative designs administering a placebo in the study.

Introduction

In the crossover design, all treatments being tested are applied to all subjects, while in the parallel group design, each subject is assigned to one specific treatment group and only undergoes that treatment. In addition to the advantage of greater power, there is generally less expense associated with measuring more responses per subject and using fewer subjects.

A criticism of the crossover design is that the effect of a treatment on a subject may be influenced by previous treatment in several possible ways: The previous treatment may continue to work; it may enhance or cancel out the effect of the current treatment; or it may cure the disorder being treated to such a degree that the current treatment does not have enough disorder to work on. Another complication with the crossover design arises when some responses are missing from the data due to subjects dropping out or failing to report at certain times; this is a less serious concern with the parallel design. Fleiss [2] notes that the FDA generally views crossover designs unfavorably. Nevertheless, the above concerns can be addressed by minimizing the effects of previous treatments (accomplished by instituting a "washout period" between treatments) and by accounting for such effects in the statistical analysis of the data.

It should also be noted that in some studies, treatments are applied simultaneously in such a way that they do not interfere with one another (for example, in a study of skin treatments, patches are placed on different areas of the skin at the same time). While these designs are similar to the sequential crossover design, they are not called crossover, but rather "split-plot", "split-mouth", "split-arm", etc. Our attention will be strictly limited to the sequential crossover study.

Model and Design

There are g treatments to be tested in a crossover study, and therefore, each subject will require g time periods to complete them. Depending on the nature of the treatments, there may be a "washout" period in between treatment periods. Any washout period would not affect the design, so we will use the term **period** to mean treatment period, that is, the numerical order in which a treatment is applied (first, second, third, etc.). We will use the term **carryover** (also called "residual") to mean the treatment applied in the previous period. We will label the treatments T_1, T_2, K , T_g and the periods P_1, P_2, K , P_g . The time periods will be concurrent over all subjects, i.e. P_1 for subject 1 will be at the same time as P_1 for subjects 2, 3, etc.

Each observed response (the dependent variable) will have corresponding values of the independent variables: subject (random effect), treatment (sometimes referred to as "direct" effect), period, and carryover. The levels of carryover include T_1, T_2, K , T_g , but when a treatment is administered in P_1 , there is no treatment being carried over, so we must have another level of carryover (a "null" level), which we will call T_0 , so there are g+1 levels of carryover.

We assume that a response may possibly be affected by the treatment in the previous period, but not by any treatment two or more periods ago. We also consider period as an effect on the response because sometimes, there are slight instrumentation differences from day to day, or the subjects may behave differently (following instructions, etc.) as time goes on.

When T_i is administered to subject n in period P_j , with treatment T_k carried over, we model the response as described in Fleiss [2, ch.10]:

$$\text{RESPONSE} \ = \ \zeta_n + \tau_i^{} + \pi_j^{} + \kappa_k^{} + \epsilon_{n,j}^{}.$$

The parameter τ_i is the direct effect of T_i ; π_j is the period effect of P_j ; κ_k is the carryover effect of T_k ; ζ_n is an adjustment for subject n (a random effect); and $\varepsilon_{n,j}$ is the error term (with the traditional assumptions). We will be interested in estimating τ_1, τ_2, K , τ_g and testing differences between various pairs. We will also be interested in estimating other parameters and differences.

The experiment should be designed so that a reasonable balance is achieved between each pair of fixed effects (treatment, period, and carryover). A special Latin square design is described in [1] and [2] that will balance these three fixed effects in the following ways:

- Treatment balanced with period: In each period, the treatments T₁, T₂,,..., T_g are applied to an equal number of subjects.
- 2. Treatment balanced with carryover: For any treatment T_i , the other g-1 treatments $T_1, K, T_{i-1}, T_{i+1}, \ldots, T_g$ are carried over the same number of times (but T_0 is carried over more frequently).
- 3. Carryover balanced over periods P_2, P_3, K , P_g : For i = 1, 2, ..., g, treatment T_i is carried over the same number of times to period P_2 as to period P_3 , P_4 , etc..

In this family of designs, the number of subjects must be a multiple of g, or a multiple of 2g if g is odd. The basic Latin squares for g=3 and g=4 are shown below.

Table I: g = 3

SUBJECT	\mathbf{P}_{1}	P_2	P_3
1	T_1	T_2	T ₃
2	T_2	T_3	T_1
3	T ₃	T_1	T ₂
4	T_1	T ₃	T ₂
5	T_2	T_1	T_3
6	T_3	T_2	T_1

Table II: g = 4

SUBJECT	P_1	P_2	P_3	P_4
1	T_1	T_2	T ₃	T_4
2	T_2	T_4	T_1	T_3
3	T_3	T_1	T_4	T_2
4	T_4	T_3	T_2	T_1

Data Analysis

To demonstrate how the response data can be analyzed using SAS, we consider an example from Fleiss [2, p.283]. In this example, four dental treatments are compared for their effects on dental plaque, and a special plaque score is recorded as the response. There are five "blocks" of four subjects each, with the treatment assignments in each block arranged in a Latin square similar to the one shown in Table II above. The treatments and responses are shown in Table III.

The SAS data set must be arranged so that there is one observation for each response. This is accomplished with the data step in Program I. Next, we run a PROC GLM to analyze the data. The parameters τ_1 , τ_2 , τ_3 , τ_4 are called non-estimable because there is no linear combination of the responses whose expected value is τ_1 , τ_2 , τ_3 , or τ_4 .

Table III: Plaque scores for 20 subjects in a four-period crossover study (Treatment # / Response Plaque Score)

	P_1	P_2	P_3	P_4	
SUB	Tr. Res.	Tr. Res.	Tr. Res.	Tr. Res.	
1	3 86	4 77	2 102	1 122	
2	4 79	1 110	3 77	2 106	
2 3	2 77	3 90	1 105	4 120	
4	1 52	2 85	4 69	3 73	
5	2 74	4 73	1 83	3 105	
6	4 96	3 97	2 94	1 113	
7	3 83	1 87	4 91	2 104	
8	1 82	2 93	3 102	4 106	
9	2 81	4 74	3 88	1 111	
10	3 64	2 78	1 77	4 76	
11	1 72	3 76	4 89	2 109	
12	4 81	1 70	2 93	3 119	
13	3 77	1 57	2 75	4 93	
14	2 76	3 70	4 60	1 107	
15	1 87	4 80	3 70	2 93	
16	4 66	2 84	1 68	3 119	
17	4 87	2 84	3 104	1 122	
18	1 72	3 83	2 91	4 92	
19	2 69	1 65	4 77	3 100	
20	3 81	4 86	1 61	2 122	

Source: Varma, A.O., Fertig, J.W., Chilton, N.W., and Mandel, I.D. *Restricted Latin Square Design in a Plaque Disclosant Study*.

Journal of Periodontal Research 9 (1974) p.387

The reason for this is as follows: The expectation of any linear combination of the responses is of the form

$$a_1\zeta_1 + ... + a_{20}\zeta_{20} + b_1\tau_1 + ... + b_4\tau_4$$

+ $c_1\pi_1 + ... + c_4\pi_4 + d_0\kappa_0 + ... + d_4\kappa_4 + e\mu$.

For each individual observation,

$$a_1 + ... + a_{20} = b_1 + ... + b_4$$

= $c_1 + ... + c_4 = d_0 + ... + d_4 = e$
and $c_1 = d_0$.

Therefore, these equations hold for any linear combination of observations. An unbiased estimate

of τ_3 , for example, would correspond to $b_3 = 1$, and all other coefficients 0, which violates the above equations. We can, however, produce unbiased estimates of the following:

(1) $\tau_i - \tau_j$

(We can test equality of any two treatments.)

(2)
$$\tau_i + \overline{\zeta} + \pi_1 + \kappa_0$$

(the average effect of treatment i in period 1, with no carryover)

(3)
$$\tau_i + \overline{\zeta} + \pi_1 + .25\kappa_0 + .75\overline{\kappa}$$

(the average effect of T_i over all periods, weighted by carryover)

(4)
$$\tau_i + \overline{\zeta} + \overline{\pi} + 1.25\overline{\kappa} - .25\kappa_i$$

(the average effect of T_i over all periods and carryovers other than i; this is estimated by the mean over treatment i in the data)

Note that by calculating (3) for two treatments i and j, (1) can be calculated as the difference. The LSMEANS statement in PROC GLM is used to request least square means for the treatments. Without specifying any options, the procedure would attempt to compute an estimate of

$$\tau_i + \overline{\zeta} + \overline{\pi} + \overline{\kappa}$$
.

Since this is not estimable, no least square means would be produced. In Program I, the OBSMARGIN (OM) option [4, p.51] is exercised to produce estimates of (3).

The OM option tells the procedure to weight the levels of the independent variables (other than treatment) according to their observed frequency in the data set, instead of weighting them equally. The LSMEANS are also reproducible by ESTIMATE statements using the coefficients in (3).

Program I: Plaque Example from Fleiss

```
data plaque;
input subject
   t1 r1 t2 r2 t3 r3 t4 r4;
period=1; plq=r1; treat=t1; carry=0;
  output;
period=2; plq=r2; treat=t2; carry=t1;
  output;
period=3; plq=r3; treat=t3; carry=t2;
 output;
period=4; plg=r4; treat=t4; carry=t3;
  output;
drop t1 r1 t2 r2 t3 r3 t4 r4;
cards;
               4 77
   1
       3
         86
                       2 102
                               1 122
         79
               1 110
                       3 77
                               2 106
  20
       3 81
               4 86
                       1
                          61
                               2 122
   ;
run;
proc glm data=plaque;
 class subject period treat carry;
model plq = subject period
             treat carry;
 means treat;
 lsmeans treat / om pdiff;
 estimate 'adj mean treat 2'
   intercept 1 treat 0 1 0 0
   subject .05 .05 .05 .05 .05
       .05 .05 .05 .05 .05 .05
       .05 .05 .05 .05 .05 .05
   period .25 .25 .25 .25
   carry .25 .1875 .1875 .1875;
 estimate 'treat 2 - treat 3'
   treat 0 1 -1 0;
run;
```

The adjusted means produced by Program I (see Output I) agree with formula given in Fleiss [2, (10.48), p.284]. The computation of the sum of squares table [2, p.285] agrees with our type I table and the adjusted sum of squares for the treatment effect agrees with that shown in our type III table. It should be noted that the degrees of freedom are not the same in the type III table as in the type I table because of the confounding between level 1 of period and level 0 of carryover.

Output I: Sums of Squares and Means

Output 1.	Danis of Equates and Means
Source	DF Sum of Squares
Model	28 16449.2775
Error	51 5584.2100

Total		79	22033	3.4875	
Source		DF	Туре	e I SS	F
SUBJECT PERIOD TREAT CARRY		19 3 3 3	5977 9814 397 259	7.2375 1.7375 7.9375 9.3650	2.87 29.88 1.21 0.79
Source		DF	Туре	III SS	F
SUBJECT PERIOD TREAT CARRY		2	6141 7273 426 259	3.2333 5.9809	33.21 1.30
Level of			PI lean		
1 2 3 4	20 20	8	36.15 39.50 38.20 33.60	13.8 15.9	2 4
	Least	Squa	ares Mea	ns	
	TREAT	PI	Q LSMEA	AN	
	1 2 3 4	89 87	7.052500 9.477500 7.977500 2.942500	00	
Paramete	er		Esti	mate	
adj mean treat 2 89.4775000 treat 2 - treat 3 1.5000000					

Reparameterization

It is possible to reparameterize the data and run an analysis of variance so that the type I and type III degrees of freedom will be the same for all effects. In the analysis in Program I, the difference of a degree of freedom was due to the fact that the 3 d.f. for period and the 4 d.f. for carryover overlap in 1 d.f.; therefore, they will have a total of just 6 instead of 7 d.f. in the type I table. The alternative parameterization shown in Program II isolates the shared degree of freedom and leaves period with 2 and carryover with 3 d.f. (see Output II). Here, LSMEANS will work with the OM option, but if the interaction terms are replaced with nested terms, LSMEANS will not work (although ANOVA tables are the same).

Program II: A Different Parameterization of the Data

```
data plaque; set plaque;
  if period = 1 then prdone = 'YES';
  else prdone = 'NO';
run;

proc glm data=plaque;
class subject prdone period
    treat carry;
model y = subject prdone
    period*prdone treat carry*prdone;
run;
```

Output II: ANOVA Tables

Source	DF	Sum of Squares	
Model Error Total	28 51 79	16449.2775 5584.2100 22033.4875	
Source	DF	Type I SS	F
SUBJECT PRDONE PERIOD*PRDONE TREAT CARRY*PRDONE	19 1 2 3 3	5977.2375 2541.5042 7273.2333 397.9375 259.3650	23.21 33.21
Source	DF	Type III SS	F
SUBJECT PRDONE PERIOD TREAT CARRY	19 1 2 3 3	6141.9264 2541.5042 7273.2333 426.9809 259.3650	23.21

Example with Squares as a Fixed Effect

Another example, found in Cochran and Cox [1, p.135], is a study of milk production where g = 3 and the subjects are six cows. What distinguishes their example from the one we just analyzed is that periods 1, 2, and 3 in the first square are not considered equivalent to periods 1, 2, and 3 in the second square. Moreover, a difference between the first and second squares can be a fixed effect. This example is analyzed in Littell, Freund, and Spector [3, p.205] using PROC GLM. They obtain the correct adjusted means by creating two continuous variables RESIDA and RESIDB show in Program III. However, the individual tests of these variables are not really of interest.

Program III: Milk Production Example Handled by Littell et al.

```
data cows;
  input cow square t1 $ m1
          t2 $ m2
                    t3 $ m3;
  period=1; milk=m1; treat=t1;
             carry='0'; output;
  period=2; milk=m2; treat=t2;
              carry=t1; output;
  period=3;
             milk=m3; treat=t3;
              carry=t2; output;
  drop t1 m1 t2 m2 t3 m3;
  cards;
    1 1
          A 38
                  B 25
                         C 15
          в 109
                  C 86
                         A 39
                  A 72
          C 124
                         в 27
                  C 76
          A 86
                         В 46
    5 2
          в 75
                  A 35
                         C 34
          C 101
    6 2
                  B 63
                         A 1
run;
data cows; set cows;
  if carry = 'C' then do;
    resida = -1; residb = -1; end;
  else do; resida = 0; residb = 0;
  if carry = 'A' then resida = 1;
  if carry = 'B' then residb = 1;
run;
proc glm data=cows;
  class cow period treat carry
        square;
  model milk = cow period(square)
               treat resida residb;
  lsmeans treat;
run;
```

In the type III ANOVA table (Output III), the test of PERIOD(SQUARE) with four degrees of freedom is actually testing

H₀:
$$\pi_1 = \pi_2 = \pi_3$$
 for Square 1
AND
 $\pi_1 = \pi_2 = \pi_3$ for Square 2

only when there is no carryover effect. Furthermore, the difference between Square 1 and Square 2 is not tested separately, but is included in the COW effect.

Output III: Sums of Squares and Lsmeans

	-			
		Sum of		
Source	DF	Squares		
Model	13	20163.1944		
Error	4	199.2500		
Total	17	20362.4444		
Source	DF	Type I SS	F	
COW	5	5781.1111	23.21	
PERIOD(SQUARE)	4	11489.1111	57.66	
TREAT	2	2276.7778	22.85	
RESIDA	1	546.7500	10.98	
RESIDB	1	69.4444	1.39	
Source	DF	Type I SS	F	
COW	5	3817.9500	19.16	
PERIOD(SQUARE)	4	11489.1111	57.66	
TREAT	2	2854.5500	28.65	
RESIDA	1	258.6736	5.19	
RESIDB	1	69.4444		
Least Squares Means				
TREAT	MI	LK LSMEAN		
	4.0	4061111		
A		2.4861111		
B C		5.1111111 5.7361111		
	7 0	· . / J U T T T T		

In Program IV, PROC GLM produces the correct adjusted means using the OM option, without RESIDA and RESIDB. The type III test of PERIOD*SQUARE*PRDONE correctly tests

$$H_0$$
: $\pi_2 = \pi_3$ for Square 1
AND
 $\pi_2 = \pi_3$ for Square 2

in the presence of carryover effects.

Program IV: Reparameterization

```
data cows; set cows;
  if period = 1 then prdone = 'YES';
  else prdone = 'NO';
run;

proc glm data=cows;
class cow period treat
      carry square prdone;
model milk =
  square|prdone period*square*prdone
  treat cow*square carry*prdone;
lsmeans treat / OM;
run;
```

Output IV: Sums of Squares and Lsmeans

Source	DF	Sum of Squares		
Source	Dr	squares		
	13			
	4	199.2500		
Total	17	20362.4444		
Source	DF	Type I SS	F	
SQUARE	1	18.00000	0.36	
PRDONE	1	8311.36111	166.85	
SQUARE*PRDONE	1	2.25000		
PER*SQ*PRDONE	2	3175.50000	31.87	
TREAT	2	2276.77778		
COW*SQUARE	4	5763.11111		
CARRY*PRDONE	2	616.19444	6.19	
Source	DF	Type III SS	F	
SQUARE	1	19.85294	0.40	
PRDONE	1	8311.36111	166.85	
SQUARE*PRDONE	1	2.25000	0.05	
PER*SQ*PRDONE	2	3175.50000	31.87	
TREAT	2	2854.55000	28.65	
COW*SQUARE	4	3817.95000	19.16	
CARRY*PRDONE	2	616.19444	6.19	
Least Squares Means				
TREAT	MI	LK LSMEAN		
A		.4861111		
В		.1111111		
С	76	.7361111		

Period 1 vs. Carryover 0

In some studies, it may be of interest to estimate the difference between Period 1 and the other periods, and the difference between Carryover 0 and the carryover of the treatments. The designs in the previous examples do not permit these differences to be estimated because they confound Period 1 with Carryover 0. However, if a placebo is used in the study, its carryover could possibly be considered equivalent to no carryover. Turning back to the plaque example, if treatment 4 is a placebo, the data could be reanalyzed with Program V.

Program V: Treatment 4 as a Placebo

Output V: ANOVA Tables

		Sum of	
Source	DF	Squares	
	0.0	16440 00000	
Model	28	16449.27750	
Error	51	5584.21000	
Total	79	22033.48750	
Source	DF	Type I SS	ਸ
Source	DF	Type I ss	Г
SUBJECT	19	5977.237500	2.87
PERIOD	3	9814.737500	29.88
TREAT	3	397.937500	1.21
CARRY	3	259.365000	0.79
Source	DF	Type III SS	F
SUBJECT	19	6141.926364	2.95
PERIOD	3	8124.940000	24.73
TREAT	3	426.980909	1.30
CARRY	3	259.365000	0.79

Conclusions

There are several methods to obtain adjusted treatment means in crossover studies where a carryover effect is present. Some of them involve mathematical tricks and may produce tests in the ANOVA table which are not meaningful.

The OM option in the LSMEANS statement often provides an easier, more direct way of obtaining the correct adjusted means. A more detailed MODEL statement can help do more appropriate testing. Variations of the assumptions such as with a placebo in the study or with SQUARE as a fixed effect can be handled by slightly modifying these methods.

References

- 1. Cochran, W.G. and Cox, G.M. (1957), Experimental Designs, Second Edition, New York: John Wiley & Sons, Inc.
- 2. Fleiss, J.L. (1986), *The Design and Analysis of Clinical Experiments*, New York: John Wiley & Sons, Inc.
- 3. Littell, R.C., Freund, R.J., Spector, P.C. (1991), *SAS System for Linear Models*, Third Edition, Cary, NC: SAS Institute Inc.
- 4. SAS Institute, Inc. (1996), SAS/STAT Software: Changes and Enhancements for Release 6.12, Cary, NC: SAS Institute Inc.

Acknowledgments

SAS and SAS/STAT are registered trademarks or trademarks of SAS Institute, Inc. in the USA and other countries. ® indicates USA registration.

Author Information

Mitchell L. Kotler Colgate-Palmolive Company 909 River Road Piscataway, NJ 08855 (732) 878-6095 email: mitchell kotler@colpal.com

Larry L. Laster School of Veterinary Medicine University of Pennsylvania Philadelphia, PA (215) 898-5076 email: larryl@pobox.upenn.edu